Quick Basic

Introducción

Lo habitual en los BASIC antiguos era enumerar las líneas de 10 en 10. Esto permite que podamos insertar nuevas líneas entre las ya existentes, utilizando un número intermedio. Por ejemplo, la línea 15 se ejecutaría después que la 10 y antes que la 20. A su vez, si añadimos una línea 12, ésta se pondría en funcionamiento después de la 10 y antes que la 15. 

En los BASIC modernos, como el QBasic, tenemos un editor en el que vamos tecleando las órdenes una tras otra, sin necesidad de números de línea (aunque lo permite, pero no es necesario).

Algunos BASIC antiguos obligaban a indicar el final del programa con la palabra END. En QBasic no es necesario, se sobreentiende que el programa acaba cuando no existan más líneas, de modo que esta orden sería opcional.

	BASIC antiguos
	QBasic

	 10 PRINT "Hola"
 20 PRINT "Adiós"
 30 END
	 PRINT "Hola"
 PRINT "Adiós"


Veamos el método que se utiliza en la ayuda que viene con QBasic para describir la sintaxis de una sentencia:

DO [ { WHILE | UNTIL } condición ]
[bloque de instrucciones]
LOOP
· Las palabras en mayúsculas son palabras reservadas del lenguaje QBasic.
· Los corchetes ([ ]) indican, que lo que hay dentro de estos es opcional.
· La barrita ( | ) y las llaves ({ }) significan, que podemos utilizar solo una de las ordenes que van entre las llaves y que van separadas por la barrita. No pueden ir dos de estas sentencias juntas. 

Cuando escribes las sentencias en QBasic, puedes hacerlo en minúsculas, él automáticamente las cambiara a mayúscula. Él también hace muchas cosas automáticamente, por ejemplo: 

Escribe esto en minúsculas y pegado, veras que el QBasic automáticamente va a ordenar la línea.

cls:print"hola":print"adios"
De todas maneras, tienes que estar alerta, porque puede ser que no haya cambiado algo

2) Sentencias básicas
CLS, REM y PRINT: 

Escribe en QBasic el siguiente programa y luego presiona F5 para correrlo:

REM Fecha: 01/03/2001
CLS
PRINT "Bienvenido a QBasic"
Este programa lo que hace es mostrarnos en pantalla el texto Bienvenido a QBasic. Ahora, lo que hace la sentencia REM es insertar un comentario en nuestro programa. El comentario no será tomando en cuenta por QBasic cuando se corra el programa, solo sirve como referencia para saber, por ejemplo: cuando se hizo, quien lo hizo, para que sirve el programa, para que funciona cada línea, etc...

También puedes usar ( ' ) como REM.

Ejemplo:

CLS
PRINT "Bienvenido a QBasic" 'se escribirá en pantalla Bienvenido a QBasic
Como puedes ver REM y el símbolo ( ' ) es un comando para comentar tus programas y que otros usuarios lo entiendan.

La sentencia CLS, nos permite borrar la pantalla antes de correr el programa. Es primordial limpiar la pantalla cada vez que iniciemos un programa, para que no se vean líneas de programas anteriores.

La sentencia PRINT es la que escribe datos en la pantalla. El texto que queremos que salga en pantalla debe estar entre comillas.

Sintaxis: PRINT " texto"
Puedes usar PRINT tantas veces como quieras en tu programa. 

Ejemplo: 

CLS
PRINT "Hola"
PRINT "Bienvenido a QBasic"
Si nos interesa que los mensajes aparezcan en una misma línea, justo uno detrás del otro, podemos conseguirlo escribiendo un punto y coma ( ; ) después de la primera orden PRINT. Esto hace que el cursor no avance a la siguiente línea. 

Ejemplo:

CLS
PRINT "Hola ";
PRINT "Bienvenido a QBasic"
Si observas bien, hay un espacio entre el texto Hola y la ultima comilla. Esto es para que los dos textos que saldrán en pantalla cuando corras el programa, no salgan pegados.

Utilizando una coma ( , ) en vez de un punto y coma, lo que conseguiremos es una separación de 10 espacios.

Ejemplo:

CLS
PRINT "Hola",
PRINT "Bienvenido a QBasic"
Estos dos textos se pueden colocar también en una única línea PRINT, así: 

CLS
PRINT "Hola "; "Bienvenido a QBasic"
La orden PRINT se puede abreviar por una interrogación ( ? ), así:

CLS
? "Hola"
QBasic convertiría automáticamente la segunda línea a ésta: 

CLS
PRINT "Hola"
Para mas información escribe PRINT y luego presiona F1.

TAB: 
El modificador TAB va después de PRINT y nos permite colocar texto en una determinada posición de la línea, cuando se corra el programa.

Sintaxis: TAB (n) 
n es la posición donde queremos que se empiece a imprimir nuestro texto.

Ejemplo:

CLS
PRINT TAB(5); "Hola"
PRINT TAB(25); "Bienvenido a QBasic"
Aquí el texto Hola se imprimirá después de 5 espacios, y el texto Bienvenido a QBasic en la línea de abajo y después de 25 espacios.

Para mas información escribe TAB y luego presiona F1.

LOCATE: 

Se usa para situar el texto en cualquier lugar específico de la pantalla.

Sintaxis: LOCATE (fila), (columna)
Ejemplo: 

CLS
LOCATE 5, 10 : PRINT "estamos en fila 5, columna 10"
Hace mover el comando PRINT, 5 filas más abajo y desplazada 10 caracteres.
Los dos puntos sirven para separar dos o mas sentencias en una misma línea.

El comando LOCATE no tiene que tener un valor representando la fila y la columna todo el tiempo.

Ejemplo: 

CLS
LOCATE , 15 
PRINT "estamos en columna 15"
Para mas información escribe LOCATE y luego presiona F1.

3) Variables
Variables: 
Las variables son muy importantes en todos los programas. Con las variables podrás guardar información relevante con la que podrás operar más tarde. 

Una variable es simplemente un nombre que das a un área de la memoria en la cual el valor dato es guardado. Cuando necesites sacar esta parte del dato, o modificar este valor, puedes referirte a la localización en memoria por el nombre de la variable. 

Eres libre de dar a la variable un nombre descriptivo. Por ejemplo el nombre de variable telefono es obvio que contendrá algún tipo de número de teléfono. 

El nombre de la variable debe ser: 

· El primer carácter deberá ser siempre una letra: mayúscula o minúscula.

· Después de la primera letra en el nombre de la variable podrá ser letras, números.

· El carácter final puede ser por ejemplo: %,&,!,#,$. 

· El nombre de la variable no puede ser una sentencia, por ejemplo PRINT.

· QBasic sólo permite nombres de variables no superiores a 40 caracteres.

· No puedes usar letras acentuadas, espacios, la letra ñ ni el guión de abajo ( _ ).

Tipos de Datos: 

Los tipos de datos de la variable son los siguientes:

	Declaración
	Nombre
	Descripción
	Espacio memoria

	%
	 Integer
	 Entero
	 2 bytes (16 bits)

	&
	 Long
	 Entero largo
	 4 bytes (32 bits)

	!
	 Single
	 Precisión Sencilla
	 4 bytes (32 bits)

	#
	 Double
	 Precisión Doble
	 8 bytes (64 bits)

	$
	 String
	 Cadena
	 1 byte p/caracter


Para declarar una variable de un tipo de dato, basta con poner el identificador al final. Todas las variables que utilizamos sin declarar el tipo de dato, se asumen como Single. 

Una variable que vaya a almacenar una dirección de una persona, la declaramos de la siguiente forma: 

direccion$ = "Urb. Terracota, calle 1"
Cuando guardamos una cadena de texto en una variable de tipo String, el texto debe de ir entre comillas.

Una variable de simple precisión Single, la declaramos de la siguiente forma:

numero! = 5.123E24
Una de doble precisión Double, así:

numero# = 1.45877D-298
Para declarar un exponente en una variable Single utilizamos E y para declararlo en una variable de tipo Double utilizamos D.

Ejemplo de tipos de datos:

cadena$ = "Soy una cadena"
entero% = 1200
enterolargo& = 198108
simple! = 1.5688
doble# = 3.458D+301
CLS
PRINT cadena$
PRINT "Soy un entero:"; entero%
PRINT "Soy un entero largo:"; enterolargo&
PRINT "Soy un simple:"; simple!
PRINT "Soy un doble:"; doble#
Para mas información escribe SINGLE y luego presiona F1.

DIM: 

Especifica un tipo de datos para una variable. 

Ejemplo:

DIM cadena AS STRING
cadena = "Soy una cadena"
CLS
PRINT cadena$
La orden DIM permite una mejor manera de declarar variables y evita tener que poner los identificadores de tipo cada vez que las utilicemos, lo cual ahorra una muy tediosa tarea en programas largos.

Para mas información escribe DIM y luego presiona F1.

4) Operaciones básicas
Operaciones Básicas: 

Operadores Matemáticos:
	Operador
	Descripción

	+
	 Suma

	-
	 Resta

	*
	 Multiplicación

	/
	 División

	\
	 División de enteros

	^
	 Exponente (ALT 094)

	.
	 Punto decimal


Veamos ahora como realizar las operaciones más básicas:

numero1 = 10
numero2 = 5
suma = numero1 + numero2
CLS
PRINT "La suma de"; numero1; "más"; numero2; "es igual a"; suma
Como se puede apreciar en el ejemplo anterior, hacemos una operación con 2 números (10 y 5) y guardamos el resultado en una variable (suma). Luego imprimimos en la pantalla un mensaje de la operación que realizamos, utilizando las variables numero1 y numero2. 

Cuando imprimimos variables, estas no van entre comillas, ya que si las ponemos entre comillas, QBasic las entenderá como si fueran texto e imprimirá solo el nombre de la variable. Por esto pusimos punto y coma ( ; ) para diferenciarlas del texto y además indicar que después del texto se deje un espacio para que no se amontone el número con el texto anterior o siguiente. 

Veamos este otro ejemplo:

numero1 = 5
numero2 = 25
CLS
PRINT "El primer número es igual a"; numero1
PRINT "El segundo número es igual a"; numero2
PRINT "La suma de ambos es igual a"; numero1 + numero2
Como podemos ver, también podemos realizar operaciones dentro de la misma sentencia PRINT
INPUT, CONST: 

Ya hemos visto como guardar datos y como sacarlos por la pantalla, ahora vamos a dejar que el usuario los introduzca:

Ejemplo:

CONST pi = 3.1416 'Con CONST se declara una constante cuyo valor no cambiará durante el programa
CLS 
PRINT TAB(33); "AREA DE UN CIRCULO"
PRINT : PRINT
INPUT "¿Cuál es el radio del círculo"; radio
area = pi * (radio ^ 2)
PRINT : PRINT "El área del circulo es:"; area
La sentencia INPUT se compone por:

· Un texto entre comillas (opcional), seguido de un punto y coma que le pondrá un signo de interrogación al final del texto al correrlo. También puedes colocar una coma, si no quieres que salga el signo de interrogación.
· Una o más variables donde se almacenaran valores separadas por comas.

Otro ejemplo:

CLS
PRINT TAB(33); "AREA DE UN TRIANGULO"
PRINT : PRINT
INPUT "base y la altura del triángulo (separados con una coma)", b, h
a = (b * h) / 2
PRINT : PRINT "El área del triángulo es:"; a
Ejemplos mas sencillos:

CLS
INPUT "¿Cual es tu nombre"; nombre$
INPUT "¿Cuantos años tienes"; edad
PRINT : PRINT "Hola "; nombre$; " tu edad es"; edad
Como puedes ver hay un espacio entre Hola y la comilla final, y también entre tu y la comilla inicial. Elimina esos espacios, corre el programa y veras lo que sucede. 

Otro ejemplo:

CLS
INPUT nombre$
PRINT : PRINT nombre$
En el ejemplo anterior, solo te aparecerá el signo de interrogación y esperará a que el usuario introduzcas algo y presione ENTER. La información introducida se guardará en la variable nombre$.

Para mas información escribe INPUT y luego presiona F1.

GOTO: 

El comando GOTO, lo que hace es saltar la ejecución de un programa a una línea determinada y el programa sigue normalmente desde la línea determinada hacia abajo.

Sintaxis: GOTO (etiqueta)
Ejemplo:

CLS
PRINT "Línea 2"
GOTO salto
PRINT "Línea 4" 'Esta línea no se ejecuta
salto:
PRINT : PRINT "El programa no ejecuto la línea 4"
La dirección a la que se salta se indica con una etiqueta, que es un nombre seguido por dos puntos.

Para mas información escribe GOTO y luego presiona F1.

5) Gráficos y colores
SCREEN: 

QBasic puede trabajar con diferentes modos de pantalla. Cuando digo modos de pantalla estoy hablando de diferentes tamaños y colores. Para saber los distintos tipos de resolución y colores, escribe SCREEN y presiona F1.

Ejemplo:

SCREEN 12
CLS
PRINT "la resolución de la pantalla es 640*480 con 256 Colores"
Éstos modos de pantalla son útiles para todos los usuarios de QBasic. Los modos de pantalla son muy importantes en los programas. Por ejemplo, si tu quieres dibujar líneas, cubos y círculos en la pantalla con alguna clase de método gráfico debes usar un modo de pantalla que te admita el dibujo de gráficos.

Cuando escribas un programa piensa en los colores que realmente vas a necesitar y la resolución que deba tener la pantalla. 640*480 es una resolución muy buena.

COLOR: 

La sentencia COLOR nos permite dar color al texto que saldrá por la pantalla.

Ejemplo:

SCREEN 12
CLS
COLOR 10 'Color verde
PRINT "Bienvenido a QBasic"
En este programa el texto Bienvenido a QBasic sale con un color verde. 

Sintaxis: COLOR (color de la letra), (color del fondo)
Para mas información escribe COLOR y luego presiona F1.

Coordenadas: 

Si te fijas de cerca en la pantalla del ordenador verás pequeños cuadraditos píxeles formando la imagen en la pantalla. Cada uno de esto píxeles representa una simple localización en el monitor. Para colocar un punto muy en la esquina superior izquierda de tu pantalla necesitas una coordenada. Una coordenada es un número específico de localización.

Si nos desplazamos sobre la pantalla hacia abajo, lo haremos sobre el eje y. Si nos desplazamos a lo largo de la pantalla, lo haremos sobre el eje x. 

Dentro de QBasic la esquina superior izquierda de la pantalla es la coordenada (0,0). Este valor de la coordenada representa los ejes (x,y). 

PSET: 

Este comando dibuja un punto en la pantalla.

Sintaxis: PSET (x , y) , (color del punto)
Ejemplo:

SCREEN 12
CLS
PSET (100, 100), 2
Este programa dibuja en la pantalla en las coordenadas (100,100) un punto. 100 píxeles hacia la derecha y 100 hacia abajo. El número siguiente es el color del punto situado en la pantalla. En el programa hemos utilizado un simple color de verde. Si tu no introduces el número de color cogerá por defecto el blanco. 

La resolución de la pantalla que usamos es 640 píxeles a lo largo y 480 hacia abajo. Así que puedes situar un punto en cualquier valor dentro de este rango. 

Para mas información escribe PSET y luego presiona F1.

LINE: 

El comando LINE, dibujara una línea en la pantalla.

Sintaxis: LINE (x1,y1) - (x2,y2) , (color de la línea)
Ejemplo:
SCREEN 12
CLS
LINE (100, 100) - (400, 400), 1
LINE (100, 400) - (400, 100), 2
Otro Ejemplo:

SCREEN 12
CLS
LINE (100, 100) - (200, 200), 4, B
El programa anterior dibujará un simple cuadrado. El cuadrado es creado en un color rojo. El comando LINE tiene ahora una letra B al final, que traza un rectángulo en vez de una línea. Si colocas BF en ves de B, se creara un cuadrado coloreado.

Para mas información escribe LINE y luego presiona F1.

CIRCLE: 

Sintaxis: CIRCLE (x, y) , (radio) , (color del circulo)
Ejemplo:

SCREEN 12
CLS
CIRCLE (100, 100), 50, 2
Este programa dibujará un círculo en las coordenadas (100,100) de tu pantalla con un radio de 50. 

Para mas información escribe CIRCLE y luego presiona F1.

6) If..then..else
Operadores de Relación: 

Los operadores relacionales nos sirven para determinar la relación que tiene una expresión con otra. Los operadores relacionales que manejaremos en QBasic son los siguientes:

	Operador
	Descripción

	=
	 Es igual a

	<>
	 No es igual a

	>
	 Es mayor que

	<
	 Es menor que

	>=
	 Es mayor o igual que

	<=
	 Es menor o igual que


Muchas veces se nos presentan situaciones en las que tenemos que evaluar una condición o situación. Si la condición es verdadera, entonces ejecutamos una o más instrucciones; si no es verdadera, ejecutamos otra u otras instrucciones.

Sintaxis: IF (condición) THEN (comandos) ELSE (comandos)
Veamos el siguiente ejemplo:

CLS
COLOR 2: PRINT TAB(20); "NUMERO NEGATIVO O POSITIVO"
COLOR 1: PRINT : PRINT
INPUT "Introduce un numero entero", numero
PRINT
IF numero < 0 THEN
PRINT "El número es negativo"
ELSE
PRINT "El número es positivo"
END IF
· El programa pide un número.
· Si (IF) el numero introducido es menor que cero entonces (THEN) se imprime el mensaje El número es negativo.
· Si no (ELSE), es menor que cero, entonces se imprime el mensaje El número es positivo.

Un bloque de instrucciones IF...THEN...ELSE debe terminar siempre con END IF para indicar que nuestro proceso de selección IF ha terminado.

Otro ejemplo:

CLS 
INPUT "¿Cuántos años tienes"; edad
IF edad < 18 THEN
PRINT : PRINT "Eres menor de edad"
ELSE 
PRINT : PRINT "Eres mayor de edad" 
END IF
Para mas información escribe IF y luego presiona F1.

Operadores Lógicos: 

Las condiciones suelen también ser afectadas por los Operadores Lógicos. Los Operadores Lógicos nos sirven para evaluar condiciones tales como:

Si numero1 no es igual a cero entonces... 
IF NOT numero1 = 0 THEN...
Si numero1 es igual a cero o menor que 1 entonces...
IF numero1 = 0 OR numero1 < 1 THEN... 
Si numero1 es igual a 1 y numero2 es menor que cero entonces... 
IF numero1 = 1 AND numero2 < 0 THEN…
Si numero1 es igual a cero ó numero1 es menor que cero entonces...
IF numero1 = 0 XOR numero1 < 0 THEN…
Ejemplos de los operadores lógicos:

NOT: 

Una chica le pregunta a otra:

¿Vas a ir al baile?. La otra contesta:
No, no iré.
Si no vas al baile, no verás a Fernando...
Aquí si la chica no va al baile, entonces no podrá ver a Fernando. Si va, entonces lo verá.

CLS
INPUT "¿Irá Petra al baile (si = 1, no = 0)?. Introduzca un número: ", baile
si = 1
no = 0
?
IF NOT baile = si THEN
PRINT "Petra no verá a Fernando"
ELSE
? "Petra si verá a Fernando" 
END IF
AND: 

Una persona va a sacar la cartilla y la secretaría le dice:

Necesitas el acta de nacimiento original y un comprobante de domicilio.
Para que esa persona pueda obtener su cartilla, necesita el acta y un comprobante de domicilio. Si le falta cualquiera de estos papeles entonces no se la dan.

CLS
PRINT "Introduzca la respuesta siendo SI = 1 Y NO = 0"
INPUT "¿Trajo el acta de nacimiento"; acta
INPUT "¿Trajo el comprobante de domicilio"; domicilio
si = 1
no = 0
IF acta = si AND domicilio = si THEN
PRINT "Toma tu cartilla"
ELSE 
PRINT "Vuelva cuando tenga todos los papeles"
END IF
OR: 

¿Mamá, me compras un helado?. La mamá le dice:
¿De cual quieres hijo, de mantecado ó de chocolate?. El niño le contesta:
Quiero de los dos.
Esta bien hijo.
Aquí, si el niño escoge de mantecado, de chocolate o de ambos, de todas maneras comerá helado.

CLS
INPUT "¿Quieres helado de mantecado (si = 1, no = 0)"; mantecado
INPUT "¿Quieres helado de chocolate (si = 1, no = 0)"; chocolate
si = 1
no = 0
?
IF mantecado = si OR chocolate = si THEN
PRINT "Toma tu helado"
ELSE
PRINT "No comerás helado" ' Si el niño no quisiera ninguna de las dos o de otra
END IF 
XOR: 

¿Mamá, me compras un helado?. La mamá le dice:
¿De cual quieres hijo, de mantecado ó de chocolate?. El niño le contesta:
Quiero de las dos. 
No, hijo: mantecado ó chocolate, de las dos no.
CLS
INPUT "¿Quieres helado de mantecado (si = 1, no = 0)"; mantecado
INPUT "¿Quieres helado de chocolate (si = 1, no = 0)"; chocolate
si = 1
no = 0
?
IF mantecado = si XOR chocolate = si THEN 
'Si mantecado y chocolate son si, la condición no se cumple y por lo tanto no se ejecuta 
? "Toma tu helado" 'Si el niño quiere solo alguna de las dos
ELSE
? "mantecado o chocolate, pero no de las dos o de otra"
END IF
EQV: 

Cuando aumenta la temperatura, todos los metales se dilatan:

CLS
INPUT "¿Aumentó la temperatura (si = 1, no = 0)"; temp
INPUT "¿Se dilató la varilla (si = 1, no =0)"; delta
si = 1
no = 0
?
IF temp = si EQV delta = si THEN
PRINT "Correcto"
ELSE
PRINT "Mentiroso"
END IF
IMP: 

Si estudias para el examen pasarás. 
Aquí estamos afirmando que si se estudia, entonces se pasará.

CLS
? "si = 1, no = 0"
INPUT "¿Estudiaste"; estudiar
INPUT "¿Pasaste el examen"; examen
si = 1
no = 0
?
IF estudiar = si IMP examen = si THEN
? "Si te creo"
ELSE
? "Es difícil que alguien que estudie repruebe un examen"
END IF
Para mas información sobre todo esto, escribe NOT y luego presiona F1.

IF ANIDADOS: 

Dentro de un programa podemos evaluar ciertas condiciones utilizando sentencias IF dentro de más bloques IF...THEN...ELSE. A este conjunto de sentencias agrupadas se les llama IF anidados.

Ejemplo:

CLS
INPUT "Introduce un número del 1 al 4: ", n
PRINT : PRINT "El número introducido fue: ";
COLOR 10
IF n < 3 THEN
IF n = 1 THEN PRINT "uno"
IF n = 2 THEN PRINT "dos"
ELSE
IF n = 3 THEN PRINT "tres"
IF n = 4 THEN PRINT "cuatro"
END IF
En el ejemplo podemos ver algunas cosas:

La sentencias IF pueden ejecutarse sin necesidad de que haya un ELSE después. O sea, si la sentencia se cumple la orden se ejecuta y si no, no pasa nada.

Si la sentencia IF solo tiene una orden que ejecutar, esta puede ir inmediatamente después del THEN.

El END IF solo se pone cuando existe un bloque IF...THEN..ELSE, o cuando existe un IF que si se cumple ejecutará varias instrucciones, sin que después haya un ELSE; esto es:

IF condición THEN
instrucción 1
instrucción 2
instrucción 3
...
instrucción n
END IF

Ahora veamos un ejemplo más complejo:

CLS
PRINT "PROGRAMA QUE DETERMINA SI UN CARACTER ES:"
PRINT "1) Letra minúscula de (a) hasta (c)"
PRINT "2) Letra mayúscula de (A) hasta (C)"
PRINT "3) Numero del 1 al 3"
PRINT
INPUT "Introduce un carácter: ", c$
PRINT "El carácter introducido es: ";
COLOR 2
IF c$ = "1" XOR c$ = "2" XOR c$ = "3" THEN
PRINT "número"
ELSE
IF c$ = "a" XOR c$ = "b" XOR c$ = "c" THEN
PRINT "letra minúscula"
ELSE
IF c$ = "A" XOR c$ = "B" XOR c$ = "C" THEN
PRINT "letra mayúscula"
END IF
END IF
END IF

Como el carácter introducido solo puede ser uno particular, usamos XOR.

Existen bloques IF...THEN...ELSE anidados.

Si un IF no se cumple, entonces se salta al ELSE inmediato que da lugar a otro IF.

Cada bloque IF...THEN...ELSE termina con el END IF que le queda más cerca. Por ejemplo, el ultimo bloque IF...THEN...ELSE termina con el primer END IF; luego, el bloque que contenía a este bloque, termina con el segundo END IF, y así sucesivamente.

Para mas información escribe IF y luego presiona F1.

ELSEIF: 

En pocas palabras, ELSEIF es un IF inmediatamente después de un ELSE o también lo podemos definir como un ELSE que evalúa una condición. 

Veamos el ejemplo anterior usando ELSEIF:

INPUT "Introduce un carácter: ", c$
PRINT "El carácter introducido es: ";
COLOR 2
IF c$ = "1" XOR c$ = "2" XOR c$ = "3" THEN
PRINT "número"
ELSEIF c$ = "a" XOR c$ = "b" XOR c$ = "c" THEN
PRINT "letra minúscula"
ELSEIF c$ = "A" XOR c$ = "B" XOR c$ = "C" THEN
PRINT "letra mayúscula"
ELSE
PRINT "otro carácter diferente de letra (a - c) y número (1 - 3)"
END IF 
La sentencia ELSEIF es también una parte opcional de un bloque IF...THEN...ELSE. Podemos también tener uno o varios ELSEIF dentro de un solo bloque IF...THEN...ELSE
Si el primer IF no se cumple, QBasic automáticamente va hasta el ELSEIF donde la condición se cumple y evalúa las ordenes que están después de este hasta que se topa con otro ELSEIF, entonces saldrá del bloque IF...THEN...ELSE.

Si ni el IF ni los ELSEIF se cumplen, entonces el programa se va hasta ELSE.

Otro Ejemplo:

CLS
INPUT "Número (1- 3): ", n
PRINT : COLOR 2
IF n = 1 THEN
PRINT "uno"
ELSEIF n = 2 THEN
PRINT "dos"
ELSEIF n = 3 THEN
PRINT "tres"
ELSE 
PRINT "fuera de rango"
END IF
Para mas información escribe ELSEIF y luego presiona F1.

7) Select case
El bloque SELECT CASE nos sirve para seleccionar de entre múltiples valores que pudiera tener una expresión. 

Veamos el siguiente ejemplo:

CLS
PRINT "Opción: 1"
PRINT "Opción: 2"
PRINT "Opción: 3"
PRINT
INPUT "Escoge una opción: ", opcion
PRINT
SELECT CASE opcion 
CASE IS = 1 
PRINT "Escogiste la opción: 1"
CASE IS = 2 
PRINT "Escogiste la opción: 2"
CASE IS = 3 
PRINT "Escogiste la opción: 3"
CASE ELSE 
PRINT "Escogiste una opción no válida"
END SELECT 
El ejemplo anterior nos presenta un pequeño menú con 3 opciones a escoger, luego selecciona los posibles valores que pudiera tener la variable opcion (SELECT CASE opcion); si escogemos la opción 1, y por tanto le damos a la variable opcion el valor de uno, entonces se ejecutaron las instrucciones que corresponden al caso de que la variable sea igual a 1 (CASE IS = 1), si escogemos la opción 2, entonces se ejecutarán las instrucciones que corresponden al caso de que la variable sea igual a 2 (CASE IS = 2), etc.

Dentro de cada caso que tenga la expresión podemos ejecutar una o más instrucciones, incluyendo bloques IF...THEN...ELSE y anidar bloques SELECT CASE si nos fuera necesario.

Para mas información escribe SELECT y luego presiona F1.

8) While..wend
El bloque WHILE (mientras)... WEND (seguir) nos sirve para que mientras una condición se cumpla (verdadera), se repita una serie de instrucciones, desde donde empieza el WHILE hasta que se encuentre el WEND. 

CONST pi = 3.1416
CLS
PRINT TAB(30); "AREA DE 5 CIRCULOS"
PRINT
contador = 0 'Inicializamos el contador
WHILE contador < 5 'Inicio del ciclo. Contador < 5 porque empezamos en cero
INPUT "Radio: ", radio
PRINT "El área es: "; pi * radio * radio
PRINT
contador = contador + 1 'El contador se incrementa en 1
WEND 'Seguir
En este tipo de ciclos o bucles generalmente el valor de la condición ya ha sido inicializado desde antes de entrar al ciclo (contador = 0) y se incrementa dentro del ciclo (contador = contador + 1). El ciclo se ejecutará mientras la condición sea verdadera, por lo que si desde antes de entrar al bucle la condición ya es falsa, el programa no ejecutará el ciclo.

Ejemplo de un bucle WHILE...WEND que no se ejecutará:

x = 0
CLS
WHILE x = 1 'Mientras x = 1... pero como x = 0 y no igual a 1...
PRINT "HOLA"
x = x + 1
WEND
PRINT "Fin del programa"
También podemos caer en ciclos infinitos. Esto es, ciclos que no tengan fin y terminarán nunca:

Ejemplo de un bucle WHILE...WEND infinito:

x = 1 'Inicializamos la variable
CLS
WHILE x = 1 'Mientras x = 1...
PRINT "HOLA"
WEND 
PRINT "Fin del programa"
Para poder salir de un bucle infinito, presiona las teclas CTRL + (*).

En efecto el programa anterior entrará en el ciclo ya que la condición es verdadera (x = 1), pero x siempre será igual a 1 ya que no hay nada que la modifique dentro del bucle y así poder dar una salida. 

No necesariamente tiene que ser un incremento el que nos dé la salida. Si usamos un contador, también podemos incrementarlo como queramos.

Ejemplo:

salir = 0 'Inicializamos la variable
CLS 
WHILE salir = 0 'Mientras salir = 0...
PRINT "HOLA"
INPUT "¿Quieres salir (si = 1, no = 0)"; salir
PRINT
WEND 'seguir
PRINT "Fin del programa"
Para mas información escribe WHILE y luego presiona F1.

9) Do..loop
El bloque DO (hacer)...LOOP (vuelta) tiene 4 variantes. Veamos la primera: 

DO WHILE...LOOP: 

Este tipo de bloque es muy parecido a WHILE...WEND, ya que la condición regularmente se conoce por adelantado, y existe la posibilidad de que nunca se ejecute el ciclo. Mientras la condición sea verdadera, el ciclo se ejecutará. Este tipo de bloque proporciona una forma más potente que WHILE...WEND.

Sintaxis: DO WHILE (condición)
instrucción1
instrucción2
instrucción3
[EXIT DO]
....
instrucciónn
LOOP
La sentencia EXIT DO nos permite salir de un bloque DO en cualquier momento.

Ejemplo:

x = 0 'Inicializamos la variable
CLS
PRINT "El valor de x al empezar el ciclo es"; x
DO WHILE x < 10
x = x + 1
? "Luego x vale"; x; "..."
LOOP
Cuando LOOP encuentra a la condición falsa se termina el ciclo. 

DO...LOOP WHILE: 

En este tipo de ciclos las instrucciones se ejecutarán por lo menos 1 vez, ya que la condición se evalúa cuando se termina de ejecutar el bloque de instrucciones. Se ejecutara el ciclo mientras la condición sea verdadera.

Sintaxis: DO 
instrucción1
instrucción2
instrucción3
[EXIT DO]
....
instrucciónn
LOOP WHILE (condición)
Veamos el siguiente ejemplo:

CLS
COLOR 10 'Verde brillante
PRINT TAB(20); "CALCULO DE LA VELOCIDAD DE UN VEHÍCULO"
COLOR 7 : PRINT 'Color gris normal
DO 'Hacer...
INPUT "Distancia recorrida por el vehículo (km): ", d
INPUT "Tiempo en recorrerla (h): ", t
v = d / t
PRINT "La velocidad con que viajó el vehículo es"; v; "km / h"
PRINT
INPUT "¿Desea calcular otra vez (si = 1, no = 0)"; desea
LOOP WHILE desea = 1 'Mientras desea = 1
DO UNTIL...LOOP: 

La palabra UNTIL (hasta que) nos dice que mientras la condición No sea verdadera, el bucle se ejecutará. O lo que es lo mismo, hasta que la condición sea verdadera, el ciclo se ejecutará.

Sintaxis: DO UNTIL (condición)
instrucción1
instrucción2
instrucción3
[EXIT DO]
....
instrucciónn
LOOP
Veamos el siguiente ejemplo:

CONST pi = 3.1416
CLS
COLOR 10
PRINT TAB(28); "PERIMETRO DE CIRCULOS"
COLOR 7 : PRINT : PRINT
INPUT "Cuántos perímetros va a calcular: ", n
PRINT
DO UNTIL c = n 
INPUT "Radio: ", radio
p = 2 * pi * radio
PRINT "El perímetro es:"; p
PRINT
c = c + 1
LOOP
Al crearse la variable c (contador) esta asume un valor de 0 por lo que:

· Si el número de perímetros a calcular n es 0, entonces la condición se hace verdadera antes de entrar al ciclo (en efecto 0 = 0) y por tanto el ciclo no se ejecutará.
· Si n es positiva, el ciclo se ejecutará n veces hasta que la condición sea verdadera.
· Si n es negativa, el bucle se hace infinito.

DO...LOOP UNTIL: 

En este tipo de bucles, las sentencias se ejecutarán al menos una vez y hasta que la condición sea verdadera.

Sintaxis: DO 
instrucción1
instrucción2
instrucción3
[EXIT DO]
....
instrucciónn
LOOP UNTIL (condición)
Ejemplo:

DO
'Cabecera
CLS
COLOR 10
PRINT TAB(25); "CALCULO DE LA ENERGÍA CINÉTICA"
PRINT : PRINT : COLOR 7
'Entrada de datos
INPUT "Introduzca la masa del cuerpo (kg): ", m
INPUT "Velocidad con que viaja (m / s): ", v
'Cálculo del resultado
ec = (m * v ^ 2) / 2 'Fórmula de la energía cinética
'Salida del resultado
PRINT "La energía cinética del cuerpo es de";
COLOR 15 : PRINT ec; 
COLOR 7 : PRINT "Joules"
'Ver si otro cálculo
PRINT
INPUT "Otro cálculo (si = 1, no = 0)", otro
LOOP UNTIL otro = 0
Para mas información escribe DO y luego presiona F1.

10) For..next
Sintaxis: FOR contador = valor inicial TO valor final [STEP incremento]
sentencia
[EXIT FOR ]
NEXT [contador] 

La sentencias se ejecutan hasta que el contador llegue a su valor final. Si omitimos la palabra STEP el contador se incrementa por default en 1. EXIT FOR sirve para salir del bloque FOR en cualquier momento. 

Veamos el siguiente ejemplo:

CLS
COLOR 10
PRINT TAB(30); "ÁREA DE TRIÁNGULOS"
PRINT : COLOR 7
INPUT "CUÁNTAS ÁREAS DESEA CALCULAR: ", n
PRINT
FOR contador = 1 TO n
COLOR 7
PRINT "TRIÁNGULO #:"; : COLOR 15 : PRINT contador
INPUT "Base: ", b
INPUT "Altura: ", h
a = (b * h) / 2
PRINT "El área es"; : COLOR 15 : PRINT a
PRINT
NEXT contador
Para explicar mejor, escribe el siguiente programa tal como se te muestra abajo: 

CLS
FOR x = 1 TO 10
PRINT "Hola"
NEXT x
Te encontrarás con que la frase Hola aparece en una lista 10 veces seguidas en la pantalla. Ahora vamos a mirar como funciona este programa. 

x es una variable y tomará el valor 1. Cada vez que el comando NEXT x es alcanzado, el programa comprobará que la variable x es 10. Si no es 10 el programa regresará donde estaba el comando FOR. Siempre que la variable x no sea 10 el regresará y sumará 1 al numero que tenía. Si tú modificas la orden FOR para que llegue hasta 20, NEXT comprobará que x ha alcanzado el valor de 20. 

Por ejemplo: 

FOR x = 1 TO 10
1,2,3,4,5,6,7,8,9,10 se detiene el programa

FOR x = 1 TO 20
1,2,3,4,5,6,7,8,9,10,11,12,13,14,15,16,17,18,19,20 se detiene el programa

Si escribieses un programa como éste: 

CLS
FOR x = 1 TO 10
NEXT x
Este programa todavía hará lo mismo pero no ejecutará nada, porque no hay nada entre FOR y NEXT.

No necesariamente tenemos que contar de uno en uno. Para ello se emplea la orden STEP. Por ejemplo, podemos escribir los número pares del 1 al 20 así: 

CLS
FOR n = 2 TO 20 STEP 2 
PRINT n 
NEXT n 

O podemos contar hacia atrás: 

CLS
FOR x = 10 TO 0 STEP -1 
PRINT x 
NEXT x 

También podemos tener ciclos FOR anidados, en donde cada FOR se termina con el NEXT que le queda más cerca.

Por ejemplo, podríamos escribir la tabla de multiplicar completa así:

FOR tabla = 1 TO 10 
FOR factor = 1 TO 10 
PRINT tabla; "x"; factor; "vale:"; 
PRINT tabla * factor 
NEXT factor 
PRINT : 'línea en blanco entre tablas 
NEXT tabla 
Para mas información escribe FOR y luego presiona F1.

11) funciones matemáticas
ABS: 

Devuelve el valor absoluto de un numero. 

Ejemplo:

CLS
COLOR 2: PRINT TAB(20); "VALOR ABSOLUTO" : COLOR 7
PRINT : INPUT "Introduce un número negativo (-a): ", a
PRINT : PRINT "El valor absoluto de "; a; "es"; : COLOR 2: PRINT ABS(a) : COLOR 7
Para mas información escribe ABS y luego presiona F1.

SIN: 

Devuelve el seno de x, donde x es un ángulo dado en radianes. 

Ejemplo:

'Calcula la componente vertical de la velocidad de un misil
CONST pi = 3.1416 

CLS
COLOR 2: PRINT TAB(20); "Componente Vertical" : COLOR 7
PRINT : INPUT "Introduzca la velocidad (m / s) del misil: ", v
PRINT : INPUT "Introduzca el ángulo (en grados) del lanzamiento: ", grados
radianes = grados * pi / 180
vy = v * SIN(radianes)
PRINT : PRINT "La velocidad vertical del misil es de"; vy; "m / s"
Como son muy comunes las conversiones de grados a radianes, para hacer esto se utiliza la fórmula:

radianes = (grados * 3.1416) / 180
Para mas información escribe SIN y luego presiona F1.

COS: 

Devuelve el coseno de x, donde x es un ángulo en radianes. 

Ejemplo:

'Calcula la componente horizontal de la velocidad de un misil
CONST pi = 3.1416
CLS
COLOR 2: PRINT TAB(20); "Componente Horizontal" : COLOR 7
PRINT : INPUT "Introduzca la velocidad (m / s) del misil: ", v
PRINT : INPUT "Introduzca el ángulo (en grados) del lanzamiento: ", grados
vx = v * COS(grados * pi / 180)
PRINT : PRINT "La velocidad horizontal del misil es de"; vx; "m / s"
Para mas información escribe COS y luego presiona F1.

TAN: 

Devuelva la tangente de x, donde x es un ángulo en radianes. 

Ejemplo:

CONST pi = 3.1416
CLS
INPUT "Introduce el ángulo de una recta: ", grados
PRINT "Tangente de la recta:"; : COLOR 15: PRINT TAN(grados * pi / 180)
PRINT : COLOR 7
INPUT "Presione <ENTER> para continuar", enter$
Para mas información escribe TAN y luego presiona F1.

ATN: 

Devuelve la arcotangente de una expresión numerica, osea, la inversa de la tangente.

'Calcula el ángulo del misil a partir de los componentes de la velocidad
CONST pi = 3.1416
CLS
INPUT "Introduzca la velocidad horizontal: ", vx
INPUT "Introduzca la velocidad vertical: ", vy
'Si la componente en x de la velocidad es 0 se producirá un error ya
'que estaremos dividiendo entre 0. Esto lo puedes arreglar con un IF
v = vy / vx
radianes = ATN(v) 'ATN devolverá un resultado en RADIANES
grados = (radianes * 180) / pi
PRINT "El ángulo (en grados) del lanzamiento es:"; : COLOR 15: PRINT grados
PRINT : COLOR 7
INPUT "Presione <ENTER> para continuar", enter$
Para mas información escribe ATN y luego presiona F1.

LOG: 

Devuelve el logaritmo natural (base e) de x, y donde x es un valor mayor que cero.

CLS
PRINT "El logaritmo natural de 9 es:"; LOG(9) 
Para mas información escribe LOG y luego presiona F1.

EXP: 

Devuelve e elevado a una potencia especificada x, donde x es un valor menor o igual a 88.02969. A esta función se le llama antilogaritmo.

CLS
n = 9
PRINT "e elevado a la potencia:"; n; "da"; EXP(n)
Para mas información escribe EXP y luego presiona F1.

SWAP: 

Intercambia los valores de dos variables.

x = 5 : y = - 3
CLS
COLOR 15 : PRINT "Primero:"
COLOR 7: PRINT "x ="; x ; "y ="; y
SWAP x, y
COLOR 15 : PRINT "Luego:"
COLOR 7: PRINT "x ="; x ; "y ="; y
Para mas información escribe SWAP y luego presiona F1.

FIX: 

Le quita los decimales a x, siendo x un numero real.
La función FIX trunca un entero, no lo redondea.

CLS
n = 3.1416
PRINT "Si le quitamos los decimales a"; n; "el resultado es:"; FIX(n)
Para mas información escribe FIX y luego presiona F1.

INT: 

Redondea x al número menor inmediato que sea menor o igual a x.

n = 3.65
CLS
PRINT "Si redondeamos con INT a"; n; "el resultado es"; INT(n)
Para mas información escribe INT y luego presiona F1.

SQR: 

Calcula la raíz cuadrada de un número o expresión mayor o igual a cero.

CLS : COLOR 15
PRINT TAB(15); "CÁLCULO DE LA HIPOTENUSA A PARTIR DE LOS CATETOS"
PRINT : PRINT : COLOR 7
INPUT "Cateto opuesto: ", cop
INPUT "Cateto adyacente: ", cad
hip = SQR(cop * cop + cad * cad)
PRINT
PRINT "La hipotenusa vale"; : COLOR 15: PRINT hip
Para mas información escribe SQR y luego presiona F1.

SGN: 

Devuelve un valor indicando el signo de una expresión numerica.

La función SGN trabaja de la siguiente forma:

· Si x es negativo, SGN devuelve -1.
· Si x es cero, SGN devuelve 0.
· Si x es positivo, SGN devuelve 1.

Para mas información escribe SGN y luego presiona F1.

12) Arreglos
Arreglos de una dimensión: 

Hasta el momento hemos visto las estructuras de datos más sencillas que son las variables, pero ahora veremos otras estructuras de datos que se llaman arreglos.

Para ver más o menos para que se utilizan estas estructuras, vamos a ver lo siguiente:

Debemos introducir los nombres de 4 personas y guardarlos para luego volverlos a imprimir en el mismo orden en que entraron. La primera forma que se nos podría ocurrir, es usar 4 variables de tipo String, guardar los nombres en ellas y luego volverlos a imprimir, algo así como:

CLS
PRINT "CAPTURA DE PERSONAS:"
INPUT "Persona 1"; persona1$
INPUT "Persona 2"; persona2$
INPUT "Persona 3"; persona3$
INPUT "Persona 4"; persona4$
PRINT "SALIDA DE PERSONAS:"
PRINT "Persona 1"; persona1$
PRINT "Persona 2"; persona2$
PRINT "Persona 3"; persona3$
PRINT "Persona 4"; persona4$
Esta es la manera más trabajosa de hacerlo ¡Que tal si fueran 100 o 200 personas!

Los arreglos (también llamados arrays, matrices o vectores) son espacios de memoria que se utilizan para almacenar información de un tipo de dato determinado. Podemos tener arreglos de enteros, de cadena, de simple precisión, etc. Podemos imaginar un arreglo sencillo de la siguiente forma:

[image: image1.jpg]Arreglo sencillo de 4 elementos





Cada cuadro (o dirección del arreglo) es un espacio reservado en memoria para guardar información del tipo que vaya a ser el arreglo. Siguiendo con el ejemplo de las 4 personas, se supone que el arreglo será de tipo String ya que almacenaremos cadenas de caracteres. Para utilizar un arreglo, debemos declararlo primero con la siguiente sintaxis:

DIM nombre del arreglo(tamaño del arreglo) AS tipo de dato
Siguiendo con el ejemplo anterior, podemos declarar nuestro arreglo de la siguiente forma:

DIM personas(4) AS STRING
Ahora veremos como llenar las localidades que reservamos en el arreglo. 

DIM personas(4) AS STRING
CLS
PRINT "CAPTURA DE PERSONAS:"
INPUT "Persona 1"; persona(1)
INPUT "Persona 2"; persona(2)
INPUT "Persona 3"; persona(3)
INPUT "Persona 4"; persona(4)
PRINT "SALIDA DE PERSONAS:"
PRINT "Persona 1"; persona(1)
PRINT "Persona 2"; persona(2)
PRINT "Persona 3"; persona(3)
PRINT "Persona 4"; persona(4)
En realidad el programa anterior es muy trabajoso todavía, pero nos ilustra de una manera como es que se llenan las direcciones de un arreglo. De lo anterior podemos ver que el nombre de la persona 1 se almacenará en la dirección 1 del arreglo, la persona 2 en la dirección 2, la persona 3 en la dirección 3 del arreglo, etc, etc, etc. Si quisiéramos llenar una dirección 5 se produciría un mensaje límite fuera de rango, ya que queremos poner valores en una dirección que no existe.

Como las direcciones están especificadas por un número, podemos utilizar un FOR para rellenar el arreglo en la posición del contador, empezando el contador en el límite inferior del arreglo y terminando en el límite superior. Veamos como se hace:

DIM personas(4) AS STRING 'Declaramos nuestro arreglo
CLS
PRINT "CAPTURA DE PERSONAS:"
FOR c = 1 TO 4
PRINT "Persona"; c; ":"; : INPUT "", personas(c) 'Rellena arreglo en la posición de c
NEXT c
PRINT "SALIDA DE PERSONAS:"
FOR c = 1 TO 4
PRINT "Persona"; c; ":"; personas(c) 'Imprime arreglo en la posición c
NEXT c
Existe otra forma de declarar un arreglo que es la siguiente:

Sintaxis: DIM nombre del arreglo(límite inferior TO límite superior) AS tipo de dato
De lo anterior podemos deducir que es posible tener arreglos cuyo límite inferior sea cualquier entero corto. Por ejemplo:

DIM elementos(3 TO 19) AS INTEGER
DIM goles(0 TO 15) AS INTEGER
El limite inferior por default es 0. Sí el límite inferior es cero, porqué empezamos a rellenar el arreglo desde la posición 1?. Si consideramos el formato:

DIM nombre del arreglo(tamaño del arreglo) AS tipo de dato
podemos ampliar lo siguiente:

· Si empezamos a rellenar el arreglo con 1 como límite inferior, entonces QBasic lo entenderá y nos dejará que lo rellenemos desde 1 para terminar en 4. En pocas palabras, si empezamos a rellenar en 1, el tamaño del arreglo permanece igual.

· Si queremos llenar el arreglo empezando desde 0, entonces terminaremos de introducir todas las personas en la posición 3 del arreglo. Pero como en realidad tamaño del arreglo es el límite superior, todavía podríamos introducir el nombre de una persona más en la posición 4 del arreglo. En pocas palabras si comenzamos a rellenar el arreglo desde 0, el tamaño del arreglo se incrementaría en 1.

Ya que QBasic lo permite, es recomendable empezar a rellenar el arreglo desde 1, aunque hay quién prefiere establecer desde el principio los límites del arreglo usando:

DIM personas(1 TO 4) AS STRING
Aunque no es muy común que se utilicen arreglos cuyo límite inferior sea diferente de 1, puede que se dé el caso y por eso se usa el TO para poner donde empieza y donde termina un arreglo.

Arreglos de más de una dimensión: 

Supongamos que queremos capturar los siguientes datos de 5 personas para luego imprimirlos en el mismo orden en que entraron:

· Nombre
· Apellido
· Edad

A simple vista se puede antojar utilizar 5 (1 por cada persona) arreglos de 3 elementos (en cada dirección iría un dato) tipo String; o también 3 (1 por cada dato) arreglos de 5 elementos (en cada elemento iría una persona). Como esto es algo tedioso (imaginemos unas 100 personas y unos 30 datos para algún curriculum...), lo mejor es utilizar un arreglo de dos dimensiones. En realidad la sentencia DIM no se limita a declarar variables o arreglos de 1 sola dimensión, sino que va mucho más allá:

DIM personas(1 TO 10, 1 TO 5) ó DIM personas(10, 5) 'Arreglo de 2 dimensiones

DIM abuelos(2, 3, 15) ó
DIM abuelos(1 TO 2, 1 TO 10, 1 TO 15) 'Arreglo de 3 dimensiones
Las dimensiones de un arreglo en QBasic pueden ser desde 0 (que es una variable sencilla) hasta 60 dimensiones; y el rango de los límites puede ser cualquier número en el intervalo de -36768 a 36767, pero el número de elementos no debe sobrepasar los 36767. Ahora veremos como utilizar un arreglo de dos dimensiones:

DIM personas(5, 3) AS STRING
'Entrada de datos
FOR individuo = 1 TO 5
COLOR 10
CLS
PRINT TAB(33); "CAPTURA DE DATOS" : PRINT
COLOR 15
PRINT "DATOS DE LA PERSONA"; : PRINT individuo
COLOR 7: PRINT
FOR dato = 1 TO 3
IF dato = 1 THEN
INPUT "Introduzca el nombre ->", personas(individuo, dato)
ELSEIF dato = 2 THEN
INPUT "Introduzca el apellido ->", personas(individuo, dato)
ELSEIF dato = 3 THEN
INPUT "Introduzca el sexo (M / F ) ->", personas(individuo, dato)
END IF
NEXT dato
PRINT
COLOR 15
PRINT TAB(22); "HECHO. PRESIONE <ENTER> PARA CONTINUAR"; : INPUT "", enter$
CLS
NEXT individuo
'Salida de datos
FOR individuo = 1 TO 5
COLOR 10
CLS
PRINT TAB(33); "SALIDA DE DATOS": PRINT
COLOR 15
PRINT "DATOS DE LA PERSONA"; : PRINT individuo
COLOR 7: PRINT
FOR dato = 1 TO 3
IF dato = 1 THEN
PRINT "Nombre ->"; personas(individuo, dato)
ELSEIF dato = 2 THEN
PRINT "Apellido ->"; personas(individuo, dato)
ELSEIF dato = 3 THEN
PRINT "Sexo (M / F) ->"; personas(individuo, dato)
END IF
NEXT dato
PRINT
COLOR 15
PRINT TAB(22); "HECHO. PRESIONE <ENTER> PARA CONTINUAR"; : INPUT "", enter$
CLS
NEXT individuo
· En el ejemplo anterior utilizamos un arreglo de dos dimensiones para almacenar los datos de las personas. La primera dimensión identifica a las personas en una dirección respectiva (la persona 1 se representa en la dirección 1 de la primera dimensión del arreglo, la persona 2 en la dirección 2, la persona 3 en la 3, etc.) y en la segunda dimensión se guardan los datos de cada persona.

· Con el primer FOR nos posicionamos en la dirección de la persona y cuando con el segundo FOR terminamos de recolectar los datos, entonces se continua con la siguiente persona.

· Los datos se guardan usando la conveniencia de que el elemento 1 de la segunda dimensión corresponde al nombre de la persona, el elemento 2 al apellido y el elemento 3 a la edad. Por esto puse los ELSEIF.

Ahora veamos como utilizar un arreglo de 3 dimensiones:

'Recolecta las horas trabajadas por semana de 2 trabajadores pertenecientes
'a 3 departamentos diferentes de 2 sucursales de una cadena comercial.
DIM empleados(2, 3, 2) AS INTEGER
FOR sucursal = 1 TO 2
COLOR 15
CLS
PRINT TAB(20); "CAPTURA DE HORAS TRABAJADAS SEMANALMENTE"
PRINT
COLOR 15
PRINT TAB(30); "SUCURSAL"; sucursal
PRINT
FOR departamento = 1 TO 3
COLOR 15
PRINT "DEPARTAMENTO"; departamento
FOR empleado = 1 TO 2
COLOR 7
PRINT "Horas trabajadas por el empleado"; empleado; "->";
COLOR 10: INPUT "", empleados(sucursal, departamento, empleado)
NEXT empleado
PRINT
NEXT departamento
NEXT sucursal
· La primera dimensión del arreglo corresponde a la sucursal, la segunda parte al departamento en donde trabaja el empleado y en la 3ª dimensión se almacenará el no. de horas que trabajó en la semana.

· El primer FOR nos sirve para primeramente colocarnos en la sucursal, el segundo FOR para colocarnos en el departamento, y el 3er. FOR nos sirve para recolectar las horas trabajadas de los empleados; cuando este ultimo FOR termina, entonces se prosigue con el siguiente departamento; y cuando se terminan los departamentos, entonces continuamos con la siguiente sucursal.

LBOUND y UBOUND: 
Estas dos funciones nos permiten obtener los límites inferior y superior de la dimensión de un arreglo (UBOUND, abreviatura de Upper Bound o límite superior; LBOUND, abreviatura de Lower Bound o límite inferior) respectivamente. 

Sintaxis:
UBOUND(nombre del arreglo[,dimensión])
LBOUND(nombre del arreglo[,dimensión])
Si el arreglo es de una dimensión, entonces solo ponemos el nombre del arreglo. Por ejemplo si tenemos el arreglo:

DIM cubo(1 TO 20, 2 TO 9, 3 TO 15) AS SINGLE
Entonces: 

LBOUND(empleados, 1) 'Devolverá 1 que es el límite inferior de la primera dimensión
LBOUND(empleados, 2) 'Devolverá 2 que es el límite inferior de la segunda dimensión
LBOUND(empleados, 3) 'Devolverá 3 que es el límite inferior de la tercera dimensión
y

UBOUND(empleados, 1) 'Devolverá 20 que es el límite superior de la 1° dimensión
UBOUND(empleados, 2) 'Devolverá 9 que es el límite superior de la 2° dimensión
UBOUND(empleados, 3) 'Devolverá 15 que es el límite superior de la 3° dimensión
Si quisiéramos obtener los límites superior e inferior del siguiente arreglo:

DIM vector(3) AS INTEGER
Es posible que si en un programa empezamos a rellenar este arreglo desde 1 creamos que como empezamos a rellenar desde 1, entonces LBOUND nos retornará 1; lo cual no hay nada más falso. Por otro lado y como ya hemos visto que los arreglos comienzan virtualmente en 0, entonces:

LBOUND(vector) 'Devolverá 0 ya que los arreglos empiezan desde 0.
UBOUND(vector) 'Devolverá 3 ya que en efecto el límite superior es 3.
De lo anterior podemos deducir que:

· Suponiendo que empezamos a rellenar un arreglo desde un número que no sea cero y luego queremos referirnos al menor elemento en una determinada dimensión utilizando LBOUND, entonces deberemos poner los limites usando TO. Ejemplo:

DIM prueba(3 TO 6)
CLS
PRINT "El limite inferior del arreglo prueba es", LBOUND(prueba)
PRINT "El limite superior del arreglo prueba es", UBOUND(prueba)
· Podemos utilizar la declaración OPTION BASE n para indicar que el limite inferior de todos los arreglos de nuestro programa que no declaremos con TO empezarán en n, donde n puede ser 0 o 1. Ejemplo:

OPTION BASE 0
DIM prueba(3 TO 6)
DIM perros(3)
CLS
PRINT "El limite inferior del arreglo prueba es", LBOUND(prueba)
PRINT "El limite superior del arreglo prueba es", UBOUND(prueba)
PRINT
PRINT "El limite inferior del arreglo perros es", LBOUND(perros)
PRINT "El limite superior del arreglo perros es", UBOUND(perros)

Si se va a utilizar la declaración OPTION BASE, entonces esta debe de ir antes de cualquier arreglo. TO ofrece una mejor manera de definir los límites del arreglo.

Arreglos estáticos y dinámicos: 

Existen dos tipos de arreglos: estáticos y dinámicos. Una arreglo estático es un arreglo cuyo número de elementos en cada dimensión permanecerá igual durante le ejecución del programa; por otro lado un arreglo dinámico es un arreglo cuyo número de elementos en cada dimensión puede cambiar en el transcurso del programa; o sea, puede hacerse más chico o más grande. Hasta el momento solo hemos manejado arreglos estáticos. Ya que los arreglos ocupan espacio en memoria, debemos indicarle al compilador si queremos que haga flexible a los arreglos y a la memoria, ó indicarle que los arreglos permanecerán con el mismo número de elementos durante todo el programa. 

Si vamos a usar arreglos dinámicos en nuestro programa, entonces debemos de poner $DYNAMIC al inicio del programa y a manera de comentario, esto es:

'$DYNAMIC ó REM $DYNAMIC
Por otro lado si en nuestro programa solo vamos utilizar arreglos estáticos, entonces podemos poner $STATIC al inicio del programa, aunque no es necesario.

'$STATIC ó REM $STATIC
Todos los arreglos por default son estáticos y tipo Single.

ERASE y REDIM: 

La sentencia ERASE reinicializa un arreglo poniendo todos los elementos a cero y todas las cadenas en nulo ( "" ) (en pocas palabras lo resetea). Ejemplo:

'Ejemplo de borrado de arreglos.
DIM prueba(1 TO 5)
CLS
PRINT "le damos valores al arreglo prueba:"
FOR x = 1 TO 5 STEP 1
prueba(x) = x
PRINT "Prueba ("; x; ") ->"; prueba(x)
NEXT x
PRINT
ERASE prueba
PRINT "luego de resetear el arreglo prueba:"
FOR x = 1 TO 5 STEP 1
PRINT "Prueba ("; x; ") ->"; prueba(x)
NEXT x
Para borrar más de un arreglo solo hay que ponerlos separados por comas ( , ) después de ERASE. Por otro lado, la sentencia REDIM cambia el numero de elementos que contienen las dimensiones de un arreglo $DYNAMIC. 

'Ejemplo de arreglo dinámicos.
REM $DYNAMIC
DIM prueba(1 TO 5)
CLS
PRINT "Al principio el limite inferior del arreglo es"; LBOUND(prueba)
PRINT "y el límite superior es"; UBOUND(prueba)
PRINT
REDIM prueba(7 TO 100) 'Cambiamos lo límites del arreglo.
PRINT "Luego el limite inferior del arreglo es"; LBOUND(prueba)
PRINT "y el límite superior es"; UBOUND(prueba)
REDIM cambia el número de elementos en las dimensiones de un arreglo, no las dimensiones; o sea, no podemos tener algo como:

REM $DYNAMIC
DIM arreglo(5, 6) 'Array de 2 dimensiones
REDIM arreglo(3, 5, 6) 'Array de 3 dimensiones
Por último solo queda mencionar que al cambiar el tamaño del un arreglo, todos los elementos se resetean (números a 0 y cadenas a "" ). También es posible cambiar el número de elementos del arreglo $DYNAMIC, primero borrándolo con ERASE y luego volver a declarar los límites con DIM, aunque es más engorroso.

'Ejemplo de arreglo dinámicos II.
REM $DYNAMIC
DIM prueba(1 TO 5)
CLS
PRINT "Al principio el límite inferior del arreglo es"; LBOUND(prueba)
PRINT "y el límite superior es"; UBOUND(prueba)
PRINT
ERASE prueba 'Primero lo borramos...
DIM prueba(5 TO 20) 'y luego lo volvemos a declarar.
PRINT "Luego el limite inferior del arreglo es"; LBOUND(prueba)
PRINT "y el límite superior es"; UBOUND(prueba)
